APPENDIX - IV.

COMPUTER CODE.

APPENDIX IV.

DEFINITION OF VARIABLES.

The major arrays and symbols used in the code are defined below. Some temporary storage variables are not defined here, but their definitions are evident from the context.

AK	Assemblage stiffness matrix.
AREA	Area of triangular element.
B	Matrix (B), equation (5.35a), for quadrilateral.
BODYF	Nodal load contribution from TBODY.
BT	Matrix (B), equation (5.35a), for triangle.
C	Stress-strain matrix. (C)
CB	Matrix product (C) (B).
CF	Common factor in the computation of stress-strain matrix(C)
E	Modulus of elasticity,E. In STRESS, the strains at the element centroid.
EL	Length of element side where surface traction is prescribed.
FAC	Factor for averaging element strains.
IBAND	Semi-band width of assemblage equations, IBAND MAXBW.
IE(M. I)	Element identification array. M is element number,1 M NEL I=1,2,3,4 denote corner nodes of the element, and I=5 denotes MTYP for the element.
ISC,JSC	Nodal numbers of nodes i and j for side on which surface traction is prescribed.
ISTOP	Index used to count data errors.
I,J,K,L	Indices of the four nodes of quadrilateral in QQUAD.
I,J,K	Indices of the three nodes of triangular element in CST.
KODE(1)	Index of displacement and concentrated load conditions at node I.

KKK	In BANSOL, index designating function to be performed. KKK = 1 for triangularization of stiffness. KKK=2 for backward solution using triangularized stiffness.
MAXBW	Maximum semi-band width allowed by storage allocation declarations.
MAXDOF	Maximum degrees of freedom, MAXDOF = 2*MAXNP.
MAXEL	Maximum number of elements allowed by storage allocation declarations.
MAXMAT	Maximum number of materials allowed by storage allocation declarations.
MAXNP	Maximum number of nodes allowed by storage allocation declarations.
MAXSLC	Maximum number of surface fraction cards allowed by storage allocation declarations.
MDIM	Maximum band width MDIM = MAXBW.
MTYP	Material type number 1 MTYP NMAT.
NBODY	Option for body force. NBODY = 0 for no weight,NBODY=1 for weight force in the negative y direction.
NDIM	Maximum degrees of freedom, NDIM = MAXDOF.
NEL	Number of elements, NEL MAXEL.
NEQ	Total number of equations, NEQ NEQ = 2*NNP MAXDOF
NMAT	Number of different materials, 1 NMAT MAXMAT.
NNP	Number of nodal points, NNP MAXNP
NOLINE	In STRESS, index to limit output to 50 lines per page.
NOPT	Option for plane strain/stress.NOPT = 1 for plane strain, NOPT = 2 for plane stress.

NPROB	Problem number
NSLC	Number of surface traction cards, NSLC MAXSLC
PR	Poisson's ratio, v.
PXI) PXJ)	Nodal contributions of surface tractions in x-direction.
PYI) PYJ)	Nodal contributions of surface traction in y-direction
Q	Load vector of quadrilateral element. In STRESS, the element displacement vector.
QK	Stiffness matrix of quadrilateral element.
R	Assemblage load vector. Also computed displacements for the assemblage in MAIN, BANSOL and STRESS.
RO	Weight density of material.
SIG	Array for stresses.
SURTRX) SURTRY)	x and y components of prescribed distributed tractions.
TBODY	Total weight of triangular element.
TH	Thickness, h.
TITLE	Array for title of the problem (72 alphanumeric characters)
TK	Stiffness matrix of triangular element.
TOTALA	Total area of quadrilateral element.
U	Prescribed displacement in x or y direction in GEOMBC.
ULX(I)) VLY(I))	Concentrated load or displacement in x and y directions at node I.
X(I)) Y(I))	x and y coordinates of node I.

XQ)
) Coordinates of the nodes of a quadrilateral or triangular element.
YQ)

(XQ (5) and YQ (5) are coordinates of centroid)

1-5 GUIDE FOR DATA INPUT

Prob.No. Blank Format 3X, 9A8. One card per problem.

		TITLE OF PROBLEM		
1	5	8		80

Basic Parameters : Format 6I5,: One card per problem

NNP		NEL	NMAT	NSIC	NOPT	NBODY	
1	5	10	15	20	25	30	80

Material Properties : Format 4E 10.3. NMAT cards per problem.

E	PR	RO	TH		
1	10	20	30	40	80

Nodal Point Data: Format 2I5. 4E 10.3(see notes 3 & 4 below).

N.P.NO.	KODE	X	Y	ULX	VLY		
1	5	10	20	30	40	50	80

Element Data 6I5 Format 6I5 (see notes 5 & 6 below)

EL NO.	I	J	K	L		
1	10	15	20	25	30	80

Material Type

Surface Tractions: Format 2I5, 4E.10.3(see note 7 below)

N.P.I	N.P.J	SURTRX(I)	SURTRX(J)	SURTRY(I)	SURTRY(J)		
1	5	10	20	30	40	50	80

Exit Card: Format blank.

One Blank card at End of a Run Allows Normal EXIT from Computer	
1	80

Notes on Input Data

1. Data cards must be in proper sequence.
2. Units must be consistent.

3. Usually one card is needed for each node. However, if some nodes fall on a straight line and are equidistant, data for only the first and the last points of this group are needed. Intermediate nodal point data are automatically generated by linear interpolation.

4. Forces and/or displacements prescribed at a node are identified by KODE as explained below:

KODE	Force/Displacement Boundary Condition
0	ULX = Prescribed Load in x direction
	VLY = Prescribed Load in y direction
1	ULX = Prescribed Disp in x direction
	VLY = Prescribed Disp in y direction
2	ULX = Prescribed Load in x direction
	VLY = Prescribed Disp in y direction
3	ULX = Prescribed Disp in x direction
	VLY = Prescribed Disp in y direction

The sign of an applied force or displacement follows the sign of the coordinate directions. For instance, a force in the positive x direction is positive, and so on. For the nodes automati--cally generated as in Note 3, KODE = 0, ULX = 0 and VLY = 0 are assigned for the generated nodes.

5. IE(M,I),IE(M.2), IE(M.3) TE(M.4) denote four corner nodes, I,J,K,L, of a quadrilateral element, M. The program also permits use of triangular elements, which are indicated by repeating the third node; that is, IE(M,3) = IE(M,4), or K = L. For a right-handed coordinate system the nodes must be input counter-clockwise around the element. IE(M,5) denotes the type of material in the element.

The maximum difference between numbers of any two nodes for a given element must be less than MAXBW/2.

6. Usually one card is needed for each element. However, if some elements are on a line in such a way that their corner node indices each increase by one compared to the previous element, only the data for the first element on the line need be input. However, note that data for the last element of the assemblage must be input. For example, in Exercise 6-3, data for only elements 1,4,7,10 and 12 are needed. The omitted element data is generated internally by the computer. The same material type as the previously input element is assigned to all generated elements.

7. Surface tractions must be specified between two adjacent nodes only. The three possible cases are shown in Fig.I-1. For case (a) only SURTRX(I) and (J) are input, and columns 31-50 are left blank. For case (b) only SURTRY(I) and (J) are input and columns 11-30 are left blank. For both tractions all columns from 1 to 50 are input. For case (c) the user may need to compute the components of tractions manually. Moreover, the user must multiply all surface intensities by the thickness of the element before the intensities are input in the computer.

The signs of tractions follow the directions of coordinate axes. A traction in the negative y direction is negative, and so on.

(7)

FLOW CHART

START

↓

Read, echo print, and check data → YES

↓

Are there data errors? → YES → STOP

↓

No

↓

Compute semi-band with

↓

Band width too large? → YES → GO TO NEXT PROBLEM

↓

No

↓

For each element:

1. Further checks of data
2. Compute (C), (B), (k), (Q)
3. Condense internal nodes
4. Store (C), (B), multipliers, pivots, element centroid
5. Add (k) to (K) and (Q) to (R)

↓

Apply kinematic constaints (geometric boundary conditions)

↓

Are these data errors ? → YES → STOP

↓

Compute and print nodal displacements

↓

For each element:

1. Recover (C), (B), multipliers, pivots, element centroids
2. Recover displacements at internal nodes.
3. Compute strains, stresses, and principal stresses.
4. Print centroidal coordinates, stresses, and principal stresses

↓

GO TO NEXT PROBLEM

STOP

```
      SEND TO (ED,FORTRANPROGS)
      DUMP ON (PROGRAM SECD)
      OVERLAY PROGRAM(SECD)
      OVERLAY (1,1) DATAIN
      OVERLAY (1,2) QUAD
      OVERLAY (2,1) CST
      OVERLAY (2,2) STRESS
      OVERLAY (2,3) GEOMBC
      OVERLAY (3,1) BANSOL
      OVERLAY (3,2) ASEMBL
      INPUT 1 = CR7
      OUTPUT 2 = LP7
       CREATE 3 = MT0/(BLANKNAME000)/512
      END
      MASTER JUG
C     EXAMPLE CODE (PROGRAM) FOR PLANE STRAIN/STRESS
      DIMENSION TITLE(9)
      COMMON /B1/ NNP,NEL,NMAT,NSLC,NOPT,NBODY,MTYP,
     1E(5),PR(5),RO(5),TH(5),IE(30,5),
     2X(30),Y(30),ULX(30),VLY(30),KODE(30),ISC(5),JSC(5),
     3SURTRX(15,2),SURTRY(15,2)
      COMMON /B2/ QK(10,10),Q(10),B(3,10),C(3,3),BT(3,6),XQ(5),YQ(5)
      COMMON /B3/ IBAND,NEQ,R(60),AK(60,32)
      DATA MAXEL,MAXNP,MAXMAT,MAXBW,MAXSLC
     1    /   15,   32,     5,   32,    20/
C
C     PROBLEM IDENTIFICATION AND DESCRIPTION
C
 9999 READ(1,100)  NPROB,(TITLE(I),I=1,9)
      IF(NPROB.LE.0) GO TO 999
      WRITE(2,200) NPROB,(TITLE(I),I=1,9)
      CALL DATAIN (MAXEL,MAXNP,MAXMAT,MAXSLC,ISTOP)
      MAXDOF =2*MAXNP
C
C     COMPUTE MAX. NODAL DIFFERENCE AND SEMI-BANDWIDTH, EQ. (6-1)
      MAXDIF = 0
      DO 2 I=1,NEL
      DO 1 J=1,4
      DO 1 K=1,4
      LL = IABS(IE(I,J)-IE(I,K))
      IF(LL.GT.MAXDIF) MAXDIF=LL
    1 CONTINUE
    2 CONTINUE
      IBAND= 2*(MAXDIF +1)
      NEQ = 2*NNP
      IF(IBAND.GT.MAXBW) GO TO 900
      IF(ISTOP.GT. 0) GO TO 999
      CALL ASEMBL(ISTOP)
      IF(ISTOP.GT.0) GO TO 999
C
C     TRIANGULARIZE STIFFNESS MATRIX, EQ. (2-2), KKK=1
      CALL BANSOL(1,AK,R,NEQ,IBAND,MAXDOF,MAXBW)
C
C     SOLVE FOR DISPLACEMENTS CORRESP. TO LOAD VECTOR R, EQ.(2-3),KKK=2
      CALL BANSOL (2,AK,R,NEQ,IBAND,MAXDOF,MAXBW)
      WRITE(2,300) (I,R(2*I-1),R(2*I),I=1,NNP)
C
```

```
      CALL STRESS
C
      GO TO 9999
  900 WRITE(2,901) IBAND,MAXBW
      GO TO 9999
  100 FORMAT(I5,3X,9A8)
  200 FORMAT(1H , / 8H1PROBLEM, I5, 3H.. , 9A8/)
  300 FORMAT(1H , 37H1OUTPUT TABLE 1.. NODAL DISPLACEMENTS//
     1 13X, 4HNODE, 9X, 11HU = X-DISP., 9X, 11HV = Y-DISP./
     2 (5X,I12, 2E20.8))
  901 FORMAT(1H ,/// 12H BANDWIDTH =, I4, 25H EXCEEDS MAX, ALLOWABLE =,
     1 I4// 30H GO ON TO NEXT PROBLEM          )
  999 STOP
      END
      SUBROUTINE DATAIN(MAXEL,MAXNP,MAXMAT,MAXSLC,ISTOP)
      COMMON /B1/ NNP,NEL,NMAT,NSLC,NOPT,NBODY,MTYP,
     1E(5),PR(5),RO(5),TH(5),IE(30,5),
     2X(30),Y(30),ULX(30),VLY(30),KODE(30),ISC(5),JSC(5),
     3SURTRX(15,2),SURTRY(15,2)
C
      ISTOP=0
      READ(1,1) NNP,NEL,NMAT,NSLC,NOPT,NBODY
C
      WRITE(2,100) NNP,NEL,NMAT,NSLC,NOPT,NBODY
C
C     CHECKS TO BE SURE INPUT DATA DOES NOT EXCEED STORAGE CAPACITY
      IF(NNP.LE.MAXNP) GO TO 201
      ISTOP=ISTOP+1
      WRITE(2,251) MAXNP
  201 IF(NEL.LE.MAXEL) GO TO 202
      ISTOP = ISTOP +1
      WRITE(2,252) MAXEL
  202 IF(NMAT.LE.MAXMAT) GO TO 203
      ISTOP = ISTOP +1
      WRITE(2,253) MAXMAT
  203 IF(NSLC.LE.MAXSLC) GO TO 204
      ISTOP = ISTOP +1
      WRITE(2,254) MAXSLC
  204 IF(ISTOP.EQ.0)   GO TO 205
      WRITE(2,255) ISTOP
      STOP
C
  205 READ(1,2) (E(I),PR(I),RO(I),TH(I),I=1,NMAT)
      WRITE(2,101)
      WRITE(2,51) (I,E(I),PR(I),RO(I),TH(I),I=1,NMAT)
C
C     READ AND WRITE NODAL DATA (REF. 1)
      WRITE(2,103)
      N=1
    5 READ(1,3) M,KODE(M),X(M),Y(M),ULX(M),VLY(M)
      IF(M-N) 4,6,7
    4 WRITE(2,105) M
      WRITE(2,52) M,KODE(M),X(M),Y(M),ULX(M),VLY(M)
      ISTOP= ISTOP +1
      GO TO 5
    7 DF=M+1-N
      RX=(X(M)-X(N-1))/DF
      RY=(Y(M)-Y(N-1))/DF
    8 KODE(N) = 0
```

```
      X(N)= X(N-1) + RX
      Y(N)= Y(N-1) + RY
      ULX(N)=0.0
      VLY(N) = 0.0
    6 WRITE(2,52) N,KODE(N),X(N),Y(N),ULX(N),VLY(N)
      N=N+1
      IF(M-N) 9,6,8
    9 IF(N.LE.NNP) GO TO 5
C
C     READ AND WRITE ELEMENT PROPERTIES, TABLE 6-4
      WRITE(2,106)
   13 L=0
   14 READ(1,15)   M,(IE(M,I),I=1,5)
   16 L=L+1
      IF(M-L)117,17,18
  117 WRITE(2,118) M
      WRITE(2,53) M,(IE(M,I),I=1,5)
      ISTOP = ISTOP +1
      GO TO 14
   18 IE(L,1) = IE(L-1,1) +1
      IE(L,2) = IE(L-1,2) +1
      IE(L,3) = IE(L-1,3) +1
      IE(L,4) = IE(L-1,4) +1
      IE(L,5) = IE(L-1,5)
   17 WRITE(2,53) L, (IE(L,I),I=1,5)
      IF(M-L)20,20,16
   20 IF(NEL-L)21,21,14
   21 CONTINUE
C
C     READ  AND WRITE SURFACE LOADING (TRACTION) CARDS
      IF(NSLC.EQ.0) GO TO 31
   30 WRITE(2,108)
      DO 40 L=1,NSLC
       READ(1,41) ISC(L),JSC(L),SURTRX(L,1),SURTRX(L,2),SURTRY(L,1),
     1SURTRY(L,2)
   40 WRITE(2,42) ISC(L),JSC(L),SURTRX(L,1),SURTRX(L,2),SURTRY(L,1),
     1SURTRY(L,2)
   31 IF(ISTOP.EQ.0) GO TO 999
      WRITE(2,900) ISTOP
    1 FORMAT(6I5)
  100 FORMAT(1H , 35H INPUT TABLE 1.. BASIC PARAMETERS  //
     1         5X, 40H NUMBER OF NODEL POINTS. . . . . . . . . .,I5/
     2         5X, 40H NUMBER OF ELEMENTS. . . . . . . . . . . .,I5/
     3         5X, 40H NUMBER OF DIFFERENT MATERIALS . . . . . .,I5/
     4         5X, 40H NUMBER OF SURFACE LOAD CARDS. . . . . . .,I5/
     5         5X, 40H 1 = PLANE STRAIN, 2 = PLANE STRESS. . . .,I5/
     6         5X, 40H BODY FORCES(1 = IN-Y DIREC., 0 = NONE),I5)
  251 FORMAT(1H ,//// 33H TOO MANY NODAL POINTS, MAXIMUM =, I5)
  252 FORMAT(1H ,//// 30H TOO MANY ELEMENTS, MAXIMUM = , I5)
  253 FORMAT(1H ,//// 30H TOO MANY MATERIALS, MAXIMUM =, I5)
  254 FORMAT(1H ,//// 40H TOO MANY SURFACE LOAD CARDS, MAXIMUM = , I5)
  255 FORMAT(1H ,////28H EXECUTION HALTED BECAUSE OF, I5,13H FATAL ERROR
     1S/)
    2 FORMAT(4E10.3)
  101 FORMAT(1H , 36H INPUT TABLE 2.. MATERIAL PROPERTIES//
     1              10H  MATERIAL,5X,10HMODULUS OF,6X,9HPOISSON'S,7X,
     2 8HMATERIAL, 7X,  8HMATERIAL/
     3 4X,6HNUMBER,5X,10HELASTICITY,8X,7H  RATIO, 8X, 7HDENSITY,6X,
     4 9HTHICKNESS)
```

```
   51 FORMAT(1H , I10, 4E15.4)
  103 FORMAT(1H , 34H1INPUT TABLE 3.. NODAL POINT DATA //
     1          5X,  5HNODAL,48X, 7HX-DISP., 8X, 7HY-DISP./
     25X,5HPOINT,6X,4HTYPE,14X,1HX,14X,1HY,8X,7HOR LOAD,8X,7HOR LOAD)
    3 FORMAT(2I5,4E15.4)
  105 FORMAT(1H , 5X,17HERROR IN CARD NO., I5/)
   52 FORMAT(1H , 2I10, 4E15.4)
  106 FORMAT(1H , 34H1INPUT TABLE 4.. ELEMENT DATA     //
     1  11X, 31HGLOBAL INDICES OF ELEMENT NODES/3X,7HELEMENT,
     2   7X,1H1,7X,1H2,7X,1H3,7X,1H4,2X,8HMATERIAL)
  108 FORMAT(1H , 37H1INPUT TABLE 5.. SURFACE LOADING DATA//
     117X, 33HSURFACE LOAD INTENSITIES AT NODES/
     2 4X,6HNODE I,4X,6HNODE J,10X,2HXI,10X,2HXJ,10X,2HYI,10X,2HYJ)
   53 FORMAT(1H , I10, 4I8, I10)
   15 FORMAT(6I5)
  118 FORMAT(1H , 5X, 25HERROR IN ELEMENT CARD NO.,I5)
   41 FORMAT(2I5, 4E10.3)
   42 FORMAT(1H , 2I10, 4E12.4)
  900 FORMAT(1H ,/// 45H ASSEMBLY AND SOLUTION WILL NOT BE PERFORMED,,
     1I5, 21H FATAL CARD ERRORS   )
  999 RETURN
      END
      SUBROUTINE ASEMBL(ISTOP)
      COMMON/B1/ NNP,NEL,NMAT,NSLC,NOPT,NBODY,MYTP,
     1E(5),PR(5),RO(5),TH(5),IE(30,5),
     2X(30),Y(30),ULX(30),VLY(30),KODE(30),ISC(5),JSC(5),
     3SURTRX(15,2),SURTRY(15,2)
      COMMON/B2/ QK(10,10),Q(10),B(3,10),C(3,3),BT(3,6),XQ(5),YQ(5)
      COMMON /B3/ IBAND,NEQ,R(60),AK(60,32)
      DIMENSION LP(8)
C
C     INITIALIZE
      ISTOP= 0
      REWIND 3
C     INITIALIZE PARTS OF MATRICES C AND BT
      BT(1,4) = 0.0
      BT(1,5) = 0.0
      BT(1,6) = 0.0
      BT(2,1) = 0.0
      BT(2,2) = 0.0
      BT(2,3) = 0.0
      C(1,3)  = 0.0
      C(2,3)  = 0.0
      C(3,1)  = 0.0
      C(3,2)  = 0.0
C
C     INITIALIZE OVERALL STIFFNESS MATRIX AK AND OVERALL LOAD VECTOR R
      DO 2    I = 1,NEQ
      R(I)= 0.0
      DO 2    J = 1,IBAND
    2 AK(I,J) = 0.0
C
C     COMPUTE ELEMENT STIFFNESSES AND LOADS ONE BY ONE
      DO 10   M = 1,NEL
      IF(IE(M,5).GT.0) GO TO 11
      ISTOP = ISTOP + 1
      GO TO 10
   11 CALL QUAD(M,AREA)
      IF(AREA.GT.0.0) GO TO 16
```

```
      ISTOP = ISTOP + 1
      WRITE(2,20)M
C
C     CONDENSE ELEMENT STIFF. FROM 10*10 TO 8*8, EQ. (5-64),AND ELEMENT
C     LOADS FROM 10*1 TO 8*1, EQ. (5,64D). (REF.2)
   16 IF(IE(M,3).EQ.IE(M,4)) GO TO 26
      DO 31 J=1,2
      IJ=10-J
      IK=IJ+1
      PIVOT = QK(IK,IK)
      DO 32 K=1,IJ
      F= QK(IK,K)/PIVOT
      QK(IK,K)=F
      DO 33 I=K,IJ
      QK(I,K)= QK(I,K)- F*QK(I,IK)
   33 QK(K,I)=QK(I,K)
   32 Q(K)= Q(K)-QK(IK,K)*Q(IK)
   31 Q(IK)= Q(IK)/PIVOT
C
C     STORE MULTIPLIERS,PIVOTS,CONDENSED LOADS,STRAIN-DISP. AND STRESS-
C     STRAIN MATRICES ON SCRATCH TAPE NO. 1 (TO BE USED LATER TO COMPUTE
C     STRAINS AND STRESSES )
   26 WRITE(3)  ((QK(I,J),J=1,10),I=9,10),Q(9),Q(10),
     1((B(I,J),J=1,10),I=1,3),((C(I,J),J=1,3),I=1,3),XQ(5),YQ(5)
C
C     ASSEMBLE STIFF. AND LOADS , DIRECT STIFF. METHOD. SEC. 6-5.
C
      LIM=8
      IF(IE(M,3).EQ.IE(M,4)) LIM=6
      DO 40 I=2,LIM,2
      IJ= I/2
      LP(I-1) = 2*IE(M,IJ) - 1
   40 LP(I) = 2*IE(M,IJ)
      DO 50 LL=1,LIM
      I= LP(LL)
      R(I)= R(I) +Q(LL)
      DO 50 MM=1,LIM
      J=LP(MM)-I+1
      IF(J.LE.0) GO TO 50
      AK(I,J)= AK(I,J) + QK(LL,MM)
   50 CONTINUE
   10 CONTINUE
C
C     ADD EXTERNALLY APPL. CONC. NODAL LOADS TO R
      DO 55 N=1,NNP
      IF(KODE(N).EQ.3)GO TO 55
      K=2*N
      IF(KODE(N).EQ.1)  GO TO 57
      R(K-1) =R(K-1) + ULX(N)
      IF(KODE(N).NE.0) GO TO 55
   57 R(K)  = R(K) +VLY(N)
   55 CONTINUE
C
C     CONVERT LINEARLY VARYING SURFACE TRACTIONS TO STATIC EQUIVALENTS,
C     AND ADD TO OVERALL LOAD VECTOR R,  EQ. (6-61A).
      IF(NSLC.EQ.0) GO TO 60
      DO 61 L=1,NSLC
      I=ISC(L)
      J= JSC(L)
```

```
      II=2*I
      JJ=2*J
      DX= X(J) -X(I)
      DY= Y(J) -Y(I)
      EL= SQRT(DX*DX+DY*DY)
      PXI     = SURTRX(L,1) * EL
      PXJ     = SURTRX(L,2) * EL
      PYI     = SURTRY(L,1) * EL
      PYJ     = SURTRY(L,2) * EL
      R(II=1) = R(II-1) + PXI/3.0 + PXJ/6.0
      R(JJ=1) = R(JJ-1) + PXI/6.0 + PXJ/3.0
      R(II)   = R(II)   + PYI/3.0 + PYJ/6.0
      R(JJ)   = R(JJ)   + PYI/6.0 + PYJ/3.0
   61 CONTINUE
C     INTRODUCE KINEMATIC CONSTRAINTS (GEOMETRIC BOUNDRY CONDITIONS),
C     EQ. (6=18),  REF.1.
C
   60 DO 70   M = 1,NNP
      IF(KODE(M).GE.0.AND.KODE(M).LE.3)  GO TO 72
      ISTOP   = ISTOP + 1
      GO TO 70
   72 IF(KODE(M).EQ.0)   GO TO 70
      IF(KODE(M).EQ.2)        GO TO 71
      CALL GEOMBC(ULX(M),2*M-1)
      IF(KODE(M).EQ.1)        GO TO 70
   71 CALL GEOMBC(VLY(M),2*M)
   70 CONTINUE
      ENDFILE 3
      IF(ISTOP.EQ.0)   GO TO 81
      WRITE(2,100) ISTOP
   20 FORMAT(1H ,/ 5X, 17H AREA OF ELEMENT ,I5,14H  IS NEGATIVE /)
  100 FORMAT(1H ,//// 42H SOLUTION WILL NOT BE PERFORMED BECAUSE OF,I5,
     115H   DATA ERRORS /)
   81 RETURN
      END
      SUBROUTINE QUAD(M,TOTALA)
      COMMON /B1/ NNP,NEL,NMAT,NSLC,NOPT,NBODY,MTYP,
     1E(5),PR(5),RO(5),TH(5),IE(30,5),
     2X(30),Y(30),ULX(30),VLY(30),KODE(30),ISC(5),JSC(5),
     3SURTRX(15,2),SURTRY(15,2)
      COMMON /B2/ QK(10,10),Q(10),B(3,10),C(3,3),BT(3,6),XQ(5),YQ(5)
      COMMON /B3/ IBAND,NEQ,R(60),AK(60,32)
      I = IE(M,1)
      J = IE(M,2)
      K = IE(M,3)
      L = IE(M,4)
      MTYP = IE(M,5)
      TOTALA = 0.0
C
C     CONSTRUCT STRESS=STRAIN MATRIX C, EQ.(3=16C).  FOR PLANE STRAIN
C     NOPT=1 ,AND FOR PLAIN STRESS NOPT=2 . PRESENT CODE IS FOR ISOTROP-
C     =IC MATERIALS
      IF(NMAT.EQ.1.AND.M.GT.1) GO TO 5
             IF(NOPT.EQ.2) GO TO 2
      CF= E(MTYP)/((1.0+PR(MTYP))*(1.0=2.0*PR(MTYP)))
      C(1,1)  =  CF*(1.0-PR(MTYP))
      C(1,2)  =  CF*PR(MTYP)
      C(2,1)  =  C(1,2)
      C(2,2)  =  C(1,1)
```

```
      C(3,3) = CF*(1.0-2.0*PR(MTYP))/2.0
      GO TO 5
    2 CF=E(MTYP)/(1.0-PR(MTYP)*PR(MTYP))
      C(1,1) = CF
      C(1,2) = PR(MTYP)*CF
      C(2,1) = C(1,2)
      C(2,2) = CF
      C(3,3) = CF*(1.0-PR(MTYP))/2.0
    5 LIM = 4
      IF(K.EQ.L) LIM = 3
      XQ(5)= 0.0
      YQ(5)= 0.0
      DO 10 N=1,LIM
      NN= IE(M,N)
      XQ(N) = X(NN)
      YQ(N) = Y(NN)
      XQ(5) = XQ(5) + X(NN)/FLOAT(LIM)
   10 YQ(5) = YQ(5) + Y(NN)/FLOAT(LIM)
C
C     INITIALIZE QUAD. STIFFNESS, LOAD VECTOR AND STRAIN-DISPLACEMENT
C     VECTOR
      DO 13 II=1,10
      Q(II) = 0.0
      DO 12 JJ=1,10
   12 QK(II,JJ) = 0.0
      DO 13 JJ=1,3
   13 B(JJ,II) = 0.0
      IF(K.NE.L) GO TO 15
      CALL CST(1,2,3,TOTALA)
      GO TO 999
   15 CALL CST(1,2,5,AREA)
      TOTALA = TOTALA + AREA
      CALL CST(2,3,5,AREA)
      TOTALA = TOTALA + AREA
      CALL CST(3,4,5,AREA)
      TOTALA = TOTALA + AREA
      CALL CST(4,1,5,AREA)
      TOTALA = TOTALA + AREA
  999 RETURN
      END
      SUBROUTINE  CST(I,J,K,AREA)
      COMMON /B1/ NNP,NEL,NMAT,NSLC,NOPT,NBODY,MTYP,
     1E(5),PR(5),RO(5),TH(5),IE(30,5),
     2X(30),Y(30),ULX(30),VLY(30),KODE(30),ISC(5),JSC(5),
     3SURTRX(15,2),SURTRY(15,2)
      COMMON /B2/ QK(10,10),Q(10),B(3,10),C(3,3),BT(3,6),XQ(5),YQ(5)
      COMMON /B3/ IBAND,NEQ,R(60),AK(60,32)
      DIMENSION CB(3,6),LC(6),LT(3),TK(6,6)
C
      LT(1)    = I
      LT(2)    = J
      LT(3)    = K
C
C     COMPUTE STRAIN-DISPLACEMENT MATRIX B FOR TRIANGLE, EQ. (5-35A)
      BT(1,1) = YQ(J) - YQ(K)
      BT(1,2) = YQ(K) - YQ(I)
      BT(1,3) = YQ(I) - YQ(J)
      BT(2,4) = XQ(K) - XQ(J)
      BT(2,5) = XQ(I) - XQ(K)
```

```
      BT(2,6) = XQ(J) - XQ(I)
      BT(3,1) = BT(2,4)
      BT(3,2) = BT(2,5)
      BT(3,3) = BT(2,6)
      BT(3,4) = BT(1,1)
      BT(3,5) = BT(1,2)
      BT(3,6) = BT(1,3)
      AREA = (BT(2,4)*BT(1,3)-BT(2,6)*BT(1,1))/2.0
C
C     COMPUTE  C*B
      DO 10   II = 1,3
      DO 10   JJ = 1,6
      CB(II,JJ) = 0.0
      DO 10   KK = 1,3
   10 CB(II,JJ) = CB(II,JJ) + C(II,KK) * BT(KK,JJ)
C
C     COMPUTE (B**T)*C*B,    EQ.(5-45A)
      DO 12   II = 1,6
      DO 12   JJ = 1,6
      TK(II,JJ) = 0.0
      DO 12   KK = 1,3
   12 TK(II,JJ) = TK(II,JJ) + BT(KK,II)*CB(KK,JJ)
C
C     ADD TRIANGLE STIFFNESS TO QUADRILATERAL STIFFNESS, EQ. (6-2)
C     ADD TRIANGLE STRAIN-DISPLACEMENT MATRIX TO QUADRILATERAL STRAIN-
C     DISPLACEMENT MATRIX
      DO 15   II = 1,3
      LC(II)  = 2*LT(II)-1
   15 LC(II+3)= 2*LT(II)
      DO 30   II = 1,6
      LL       = LC(II)
      FK       = 1.0/(4.0*AREA)
      FB       = 2.0*FK
      DO 20   JJ = 1,6
      MM       = LC(JJ)
   20 QK(LL,MM) = QK(LL,MM) + TK(II,JJ) * TH(MTYP) * FK
      DO 30   JJ = 1,3
   30 B(JJ,LL) = B(JJ,LL) + BT(JJ,II)*FB
C
C     DEVELOPE BODY FORCE VECTOR, EQ.(5-61B)
      IF(NBODY.EQ.0)  GO TO 999
      TBODYF   = AREA *RO(MTYP) * TH(MTYP)
      BODYF    = - TBODYF/3.0
      DO 35   II = 1,3
      JJ       = 2*LT(II)
   35 Q(JJ)    = Q(JJ) + BODYF
  999 RETURN
      END
      SUBROUTINE  STRESS
      COMMON /B1/ NNP,NEL,NMAT,NSLC,NOPT,NBODY,MTYP,
     1E(5),PR(5),RO(5),TH(5),IE(30,5),
     2X(30),Y(30),ULX(30),VLY(30),KODE(30),ISC(5),JSC(5),
     3SURTRX(15,2),SURTRY(15,2)
      COMMON /B2/ QK(10,10),Q(10),B(3,10),C(3,3),BT(3,6),XQ(5),YQ(5)
      COMMON /B3/ IBAND,NEQ,R(60),AK(60,32)
      DIMENSION SIG(6)
C
      REWIND 3
      WRITE(2,300)
```

```
      NOLINE  = 47
C
C     RETRIEVE MULTIPLIERS,PIVOTS,MATRICES B AND C, AND CENTROIDAL COORD
C     FOR ELEMENT
8996  DO 5    M = 1,NEL
      READ (3)   ((QK(I,J),J=1,10),I=1,2),Q(9),Q(10),
     1 ((B(I,J),J=1,10),I=1,3),((C(I,J),J=1,3),I=1,3),XC,YC
C
C     SELECT NODAL DISPLACEMENT FOR THE ELEMENT
      LIM     = 4
      IF(IE(M,3).EQ.IE(M,4)) LIM = 3
      DO 10   I = 1,LIM
      II      = 2*I
      JJ      = 2*IE(M,I)
      Q(II=1) = R(JJ-1)
   10 Q(II)   = R(JJ)
C
C     RECOVER CONDENSED DISPLACEMENT FOR THE QUADRILATERAL, EQ. (5=64G)
      IF(LIM.EQ.3)  GO TO 16
      DO 15   K = 1,2
      JK      = K + 8
      IK      = JK - 1
      DO 15   L = 1,IK
   15 Q(JK)   = Q(JK) - QK(K,L)*Q(L)
C
C     COMPUTE ELEMENT STRAINS, EQ. (5=35A)
      LIM     = 10
      FAC     = 0.25
      GO TO 17
   16 LIM     = 6
      FAC  =1.00
   17 DO 20   I = 1,3
      E(I)    = 0.0
      DO 20   J = 1,LIM
   20 E(I)    = E(I) + B(I,J)*Q(J)*FAC
C
C     COMPUTE ELEMENT STRESSES, EQ.(5=35B)
      DO 30   I = 1,3
      SIG(I)  = 0.0
      DO 30   J = 1,3
   30 SIG(I)  = SIG(I) + C(I,J) *E(J)
C     COMPUTE PRINCIPAL STRESSES AND THE ANGLE WITH THE POSITIVE X AXIS
      SP      = (SIG(1) + SIG(2))/2.0
      SM      = (SIG(1) - SIG(2))/2.0
      DS      = SQRT(SM*SM+SIG(3)*SIG(3))
      SIG(4)  = SP + DS
      SIG(5)  = SP - DS
      SIG(6)  = 0.0
      IF(SIG(3).NE.0.0.AND.SM.NE.0.0) SIG(6)=28.648*ATAN2(SIG(3),SM)
C     PRINT STRESSES, 50 LINES PER PAGE
      IF(NOLINE.GT.0)  GO TO 54
      WRITE(2,1000)
      NOLINE  = 49
   54 NOLINE  = NOLINE-1
    5 WRITE(2,1010) M,XC,YC,(SIG(I),I=1,6)
      ENDFILE 3
  300 FORMAT(1H,47H1OUTPUT TABLE 2.. STRESSES AT ELEMENT CENTROIDS//
     11X,7HELEMENT,9X,1HX,9X,1HY,4X,8HSIGMA(X),4X,8HSIGMA(Y),4X,
     28HTAU(X,Y),4X,8HSIGMA(1),4X,8HSIGMA(2),7X,5HANGLE)
```

```
 1000 FORMAT(1H , 1H1,7HELEMENT, 9X,1HX,9X,1HY,4X,8HSIGMA(X),4X,8HSIGMA(
     1Y),4X,8HTAU(X,Y),4X,8HSIGMA(1),4X,8HSIGMA(2),7X,5HANGLE)
 1010 FORMAT(I8, 2F10.2, 6E12.4)
      RETURN
      END
      SUBROUTINE GEOMBC(U,N)
      COMMON /B3/ IBAND,NEQ,R(60),AK(60,32)
C     THIS SUBROUTINE MODIFIES THE ASSEMBLAGE STIFFNESS AND LOADS FOR
C     THE PRESCRIBED DISPLACEMENT U AT DEGREE OF FREEDOM N, EQ.(6-188).
C     (REF.1)
      DO 100  M = 2,IBAND
      K         = N-M+1
      IF(K.LE.0)  GO TO 50
      R(K)      = R(K) - AK(K,M)*U
      AK(K,M)   = 0.0
   50 K         = N+M-1
      IF(K.GT.NEQ) GO TO 100
      R(K)      = R(K) - AK(N,M)*U
      AK(N,M)   = 0.0
  100 CONTINUE
      AK(N,1)   = 1.0
      R(N)      = U
      RETURN
      END
      SUBROUTINE BANSOL(KKK,AK,R,NEQ,IBAND,NDIM,MDIM)
C     SYMMETRIC BAND MATRIX EQUATION SOLVER, (REF.2)
C
C     KKK = 1 TRIANGULERIZES THE BAND MATRIX AK, EQ.(2-2)
C     KKK = 2 SOLVES FOR RIGHT HAND SIDE R, SOLUTION RETURNS IN R,
C     EQ.(2-3)
C
      DIMENSION AK(60,32),R(1)
      NRS       = NEQ-1
      NR        = NEQ
      IF(KKK.EQ.2) GO TO 200
      DO 120  N = 1,NRS
      M         = N-1
      MR        = MIN0(IBAND,NR-M)
      PIVOT     = AK(N,1)
      DO 120  L = 2,MR
      CP        = AK(N,L)/PIVOT
      I         = M + L
      J         = 0
      DO 110  K = L,MR
      J         = J + 1
  110 AK(I,J)   = AK(I,J) - CP*AK(N,K)
  120 AK(N,L)   = CP
      GO TO 400
  200 DO 220  N = 1,NRS
      M         = N - 1
      MR        = MIN0(IBAND,NR-M)
      CP        = R(N)
      R(N)      = CP/AK(N,1)
      DO 220  L = 2,MR
      I         = M + L
  220 R(I)      = R(I) - AK(N,L)*CP
      R(NR)     = R(NR)/AK(NR,1)
      DO 320  I = 1,NRS
      N         = NR - I
```

```
      M        = N - 1
      MR       = MINO(IBAND,NR-M)
      DO 320  K = 2,MR
      L        = M + K
C     STORE COMPUTED DISPLACEMENTS IN LOAD VECTOR R
  320 R(N)     = R(N) - AK(N,K)*R(L)
  400 RETURN
      END
      FINISH
```